
Themenübersicht 

 Einfache Formular-Instanzen 

 Alternative zum Kommando  DoCmd.OpenForm 

 Zuweisung des Formulars in einer Objektvariable, um dieses einfach manipulieren / Daten übergeben 

können 

 Mehrfach Instanzen: Ein Formular gleichzeitig mehrfach öffnen 

 

 

 Formular-Instanzen mit Benutzerdefinierten Events 

 Auslösen von Events, worauf das aufrufende Formular reagieren und die Daten direkt 

weiterverarbeiten kann (Programmlogik bleibt im aufrufenden Formular) 

 

 

 Auslagerung von Formular-Events in globale Klassenmodule 

 Erstellung einer eigenen Klasse in welcher die Standard Formular-Events ausgelagert werden. 

Die Klasse kann von jedem Formular einheitlich verwendet werden 

Einfache Formular-Instanzen 

Klassenmodul des aufrufenden Formulars: 

Option Compare Database 

Option Explicit 

 

Dim frm As Form_MeinFormular 'Modul-Variable für das Formular  mit dem Namen MeinFormular 

    'Hinweis: Die Klasse Form_MeinFormular ist nur vorhanden, 

    'wenn das Formular mindestens eine Prozedur enthält 

 

Private Sub FormularÖffnen() 

    Set frm = New Form_MeinFormular 'Form-Instanz wird erstellt 

    'Das Form wird geladen und die üblichen vom Events Form werden ausgelöst Form_Open/Form_Load… 

     

    'Bevor das Form sichtbar ist, kann dieses manipuliert werden / können Daten übertragen werden 

    frm.Modal = True 'Das Form wird als gebunden deklariert 

    frm.Caption = "Mein Fenstertitel" 

    frm.MeineProzedur 'Eine Prozedur im Form wird ausgeführt. Diese muss im Form auf PUBLIC stehen 

     

    frm.Visible = True 'Das Form wird sichtbar geschaltet 

End Sub 

 

Private Sub Form_Close() 

    Set frm = Nothing 'Beim schließen des Formulars wird die Modul-Variable geleert 

End Sub 



Formular-Instanzen mit 

Benutzerdefinierten Events 
Klassenmodul des aufrufenden Formulars: Klassenmodul des geöffneten  Formulars: 

Option Compare Database 

Option Explicit 

 

'Modul-Variable für das Form 'MeinFormular'  

'mit benutzerdefinierten Events 

Private WithEvents frm As Form_MeinFormular 

 

Private Sub FormularÖffnen() 

    Set frm = New Form_MeinFormular 

    frm.Visible = True 

End Sub 

 

Private Sub frm_RecordUpdate(Wert As String) 

    MsgBox Wert 

End Sub 

Private Sub frm_Terminate() 

    Set frm = Nothing 

End Sub 

 

Private Sub Form_Close() 

    Set frm = Nothing 

End Sub 

Option Compare Database 

Option Explicit 

 

'2 Benutzerdefinierte Events werden definiert 

Public Event MeinEvent(Parameterwert As String) 

Public Event Terminate() 

 

Private Sub MeinEventAuslösen() 

    'Das Benutzerdefinierte Event 'MeinEvent' 

    'wird ausgelöst 

    RaiseEvent MeinEvent("Hallo") 

End Sub 

 

Private Sub Form_Close() 

    'Das Benutzerdefinierte Event 'Terminate' 

    'wird ausgelöst 

    RaiseEvent Terminate 

End Sub 



Formular-Instanzen  

Vor- und Nachteile 

Mit Formular-Instanz 

 Differenzierung zwischen öffnen und anzeigen des Formulars 
(Manipulierung des Formulars und Datenübertragung vor Anzeige) 

 Mehrfach-Instanzen 

 Auslösen von Benutzerdefinierten Events 
(Programmlogik bleibt im aufrufenden Formular) 

 Ungeeignet zur Verwendung als Abfrage-Fenster  

 Ausführung etwas umständlich (Verwendung von Objektvariablen) 

Ohne Formular-Instanz (DoCmd.OpenForm) 

 Einfache Benutzung (Keine Variable erforderlich) 

 Einfache Übergabe von Parameter 

 WindowMode acDialog zur Verwendung als Abfrage-Fenster  

 Nur einen Benutzerdefinierter Parameter (OpenArgs)  

 Keine Mehrfach-Instanzen 
 



Auslagerung von Formular-Events 

in globale Klassenmodule 

Globales Klassenmodul mit dem Namen 

clsRecordValidation: 
Klassenmodul der geöffneten  Formulare: 

Option Compare Database 

Option Explicit 

 

Private WithEvents cForm As Access.Form 

     

Public Sub SetForm(frm As Access.Form) 

    Set cForm = frm 

    'Diese Zuweisung ist zwingend erforderlich! 

    cForm.BeforeUpdate = "[Event Procedure]" 

End Sub 

        

Private Sub cForm_BeforeUpdate(Cancel As Integer) 

    If MsgBox("Datensatz jetzt speichern?" _ 

           ,vbYesNo) = vbNo then Cancel = True 

End Sub 

 

Private Sub Class_Terminate() 

    Set cForm = Nothing 

End Sub 

Option Compare Database 

Option Explicit 

 

Dim clsRV As clsFormRecordValidation 

     

Private Sub Form_Open(Cancel As Integer) 

    Set clsRV = New clsFormRecordValidation 

    clsRV.SetForm Me 

End Sub 

Aus den Augen aus den Sinn? 

 Vortrag am 17.05.2014 von 

 Jean Pierre Allain 

 Email: jpa@abiss.de  

 WebSeite: www.abiss.de 

 Blog: http://msofficeinside.blogspot.de/  

 
 

 Tools und Add-Ins: 
 Kontextmenüleisten anpassen: Shortcut Menu Manager 
 Access Linked Table Manager: ALTM 

 Duplikate suchen und entfernen 

 Darstellung und Verwaltung von zugeordnete Datensätze: Multivalued fields 

